APPENDIX

In this appendix, we simply list the subroutines RELBIL(IF) and SERPRB(IF) from VOACAP for completeness and so that the subroutine explanations in the body of this report can be followed via the actual code, if desired.

```
      SUBROUTINE RELBIL (IF)
C
C THIS ROUTINE COMPUTES THE RELIABILITY FOR EACH MODE AT A PARTICULAR
C FREQUENCY, AND SELECTS THE MOST RELIABLE, CALCULATES THE COMBINED
C DISTRIBUTION FOR ALL MODES AND THE RELIABILITY OF THE COMBINATION, AND
C THE REQUIRED POWER PLUS GAIN TO ACHIEVE THE REQUIRED RELIABILITY.  EACH
C OF THE MODE CALCULTIONS MADE IN THIS SUBROUTINE ARE BASED ON THE
C FREQUENCY POINTED TO BY IF.
C
C
C INPUT:
C         IF         =    INDEX, POINTING TO ONE OF 13 FREQUENCIES IN
C                         ARRAY FREL().
C         TLLOW(K)   =    LOWER DECILE OF TRANSMISSION LOSS DISTRIBUTION
C                         FOR MODE K (REGMOD, ESMOD).
C         TLHGH(K)   =    UPPER DECILE OF TRANSMISSION LOSS DISTRIBUTION
C                         FOR MODE K (REGMOD, ESMOD).
C         DL         =    LOWER DECILE OF THE NOISE LEVEL (GENOIS).
C         DU         =    UPPER DECILE OF THE NOISE LEVEL(GENOIS).
C         RSN        =    REQUIRED SIGNAL TO NOISE RATIO FOR THE CURRENT
C                         CIRCUIT.
C         SN(K)      =    SIGNAL TO NOISE RATIO FOR MODE K
C                         (REGMOD)(ESMOD FOR SPORADIC E LAYER).
C         HN(K)      =    NUMBER OF HOPS IN RAYPATH FOR MODE K (REGMOD,
C                         ESMOD) .
C         HP(K)      =    VIRTUAL HEIGHT OF REFLECTION FOR MODE K
C                         (RDGMOD, ESMOD).
C         B(K)       =    RADIATION ANGLE FOR THIS MODE  (REGMOD,ESMOD).
C         FLDST(K)   =    FIELD STRENGTH FOR  THIS MODE (REGMOD,ESMOD).
C         SIGPOW(K)  =    SIGNAL POWER AT THE RECEIVER FOR MODE K
C                         (REGMOD,ESMOD).
C         LAYTYP(K)  =    ARRAY OF CHARACTERS DESCRIBING THE LAYER TYPE
C                         FOR MODE K.  E,F1,F2,.... (BLOCK DATA).
C         TLOSS(K)   =    MEDIAN TRANSMISSION LOSS FOR MODE K (REGMOD,
C                         ESMOD).
C         PROB(K)    =    F DAYS IS THE PROBABILITY THAT THE CURRENT
C                         FREQUENCY WILL EXCEED THE MUF FOR MODE K
C                         (REGMOD, ESMOD).
C         TIMED(K)   =    TIME DELAY FOR MODE K (REGMOD,ESMOD).
C         RCNSE(K)   =    TOTAL NOISE AND ANTENNA EFFICIENCY AT RECEIVER
C
C
C OUTPUT:
C         RELY(K)          =    RELIABILITY FOR CURRENT FREQUENCY AND
C                               MODE K.
C         RELIAB(IF)       =    RELIABILITY OF THE SUM OF THE MODES FOR
C                               THIS FREQUENCY.
C         DBLOSL(IF)       =    LOWER DECILE OF THE FIELD STRENGTH
C                               DISTRIBUTION FOR THE SUM OF ALL MODES ON
C                               THIS FREQUENCY.
C         DBLOSU(IF)       =    UPPER DECILE OF THE FIELD STRENGTH
C                               DISTRIBUTION FOR THE SUM OF ALL MODES ON
C                               THIS FREQUENCY.
```

```
C          DBU(IF)          =    MEDIAN FIELD STRENGTH OF THE SUM OF ALL
C                                 MODES FOR THIS FREQUENCY.
C          SNDB(IF)         =    MEDIAN SIGNAL TO NOISE RATIO OF THE SUM
C                                 OF ALL MODES FOR THIS FREQUENCY.
C          NDBW(IF)         =    SIGNAL POWER OF THE SUM OF ALL MODES FOR
C                                 THIS FREQUENCY.
C          SNRLW(IF)        =    LOWER DECILE SIGNAL TO NOISE RATIO FOR
C                                 THE POWER SUM OF ALL MODES THAT GO FOR
C                                 THIS FREQUENCY.
C          SNRUP(IF)        =    UPPER DECILE OF THE SIGNAL TO NOISE RATIO
C                                 FOR THE POWER SUM OF ALL MODES THAT GO
C                                 FOR THIS FREQUENCY.
C          ANGLE(IF)        =    TAKE OFF ANGLE OF THE STRONGEST MODE THAT
C                                 GOES AT THIS FREQUENCY.
C          CPROB(IF)        =    PROBABILITY THAT THE CURRENT FREQUENCY
C                                 WILL GO (PROB() F DAYS PROBABILITY).
C          DELAY(IF)        =    TIME DELAY OF THE STRONGEST MODE FOR THIS
C                                 FREQUENCY.
C          DBLOS(IF)        =    TRANSMISSION LOSS OF THE STRONGEST MODE
C                                 FOR THIS FREQUENCY.
C          VHIGH(IF)        =    VERTUAL HEIGHT OF THE STRONGEST MODE FOR
C                                 THIS FREQUENCY.
C          MODE(IF)         =    ALPHA NUMERIC DESCRIPTION OF THE
C                                 STRONGEST MODE FOR THIS FREQUENCY E, E2,
C                                 2E, 2F2, ETC.
C          NHP(IF)          =    NUMBER OF HOPS FOR THE STRONGEST MODE FOR
C                                 THE FREQUENCY.
C          NYNOIS(IF)       =    RECEIVER NOISE.
C          SNPR(IF)         =    SIGNAL TO NOISE RATIO REQUIRED FOR THE
C                                 SPECIFIED RELIABILITY FOR SUM OF ALL
C                                 MODES AND THIS FREQUENCY.
C
C   PROCEDURE:
C
C THE RELIABILITY IS CALCULATED FOR EACH MODE OF COMMUNICATION
C THE MOST RELIABLE MODE IS SELECTED ON THE FOLLOWING BASIS:
C    1ST RELIABILITY
C    2ND NUMBER OF HOPS
C    3RD BY MEDIAN SIGNAL/NOISE RATIO
C
C AFTER THE SELECTION OF THE MOST RELIABLE MODE, THE SIGNAL POWER FOR EACH
C MODE IS POWER SUMED TO GET PATH RELIABILITY FOR THE CURRENT FREQUENCY.
C
C          SP   = SIGNAL POWER AT THE RECEIVER FOR EACH MODE
C          SU   = UPPER DECILE OF TRANSMISSION LOSS FOR EACH MODE
C          SL   = LOWER DECILE OF TRANSMISSION LOSS FOR EACH MODE
C          FS   = FIELD STRENGTH FOR EACH MODE
C
C          DSU  = UPPER DECILE SIGNAL LEVEL ADJUSTMENT FROM MEDIAN
C          ASM  = AURORAL ADJUSTMENT TO MEDIAN SIGNAL LEVEL
C          DSL  = LOWER DECILE SIGNAL LEVEL ADJUSTMENT FROM MEDIAN
C          ADS  = MEAN PREDICTION ERROR
```

```
C         SUS  = UPPER DECILE PREDICTION ERROR
C         SLS  = LOWER DECILE PREDICTION ERROR
C
C-----------------------------------------------------------------------
C              Start of Common Area
C-----------------------------------------------------------------------
C
      COMMON / CONTRL / IELECT(3), KTOUT(12), MONTHS(12), SUNSP(12),
     A IANTOU, ICARD, INTEG, IRED, ISOUT, ISPROC, ISSN, ITYPE, JDASH,
     B JFREQ, JLONG, KCARD, KRUN, MAPIN, MAXNAM, MONOLD, MOREM, MORES,
     C NUMNAM, NUPROC, MAXMET, MSPEC, M100
      COMMON / ALPHA / IMON(12), IRCVR(2), ITRAN(2), MODE(13),
     A MODER(13), IRLAT, IRLONG, ITLAT, ITLONG, NYEAR
      COMMON/ANOIS/ATNU,ATNY,CC,TM,RCNSE,DU,DL,SIGM,SXGU,SXGL,KJ,JK
      COMMON /CON /D2R, DCL, GAMA, PI, PI2, PIO2, R2D, RZ, VOFL
      COMMON /DON /ALATD, AMIN, AMIND, BTR, BTRD, DLONG, DMP, ERTR, GCD,
     1 GCDKM, PMP, PWR, RLAT, RLATD, RLONG, RLONGD, RSN, SIGTR, TLAT,
     2 TLATD,TLONG,TLONGD,BRTD,FLUX,SSN,D90R,D50R,D10R,D90S,D50S,D10S
      COMMON/FRQ/FREL(29),FREQ,JMODE,ITXRCP(2)
      COMMON / ION / IHRE, IHRO, IHRS, LUFP, METHOD, NOISE, NPSL
      COMMON / METSET / VERSN, ITRUN, ITOUT, JTRUN(40), JTOUT(40), LSEAS
      COMMON/MUFS/EMUF(24),FIMUF(24),F2MUF(24),ESMUF(24),ALLMUF(24),FOT
     A(24),XLUF(24),HPF(24),ANGMUF(24),MODMUF,SIGL(4),SIGU(4),DELMUF(4)
     B ,HPMUF(4),HTMUF(4),FVMUF(4),AFMUF(4),NHOPMF(4),YFOT(4),YHPF(4)
     C ,YMUF(4)
      COMMON / OUTLAB / LABEL(11), LAYTYP(5), IEAST, INORTH, ISOUTH,
     A IWEST, LABLI, LABLJ, LABLK
      COMMON/SIGD/DSL,ASM,DSU,AGLAT,DSLF,ASMF,DSUF,ACAV,FEAV,AFE,BFE,HNU
     A ,HTLOSS,XNUZ,XVE
      COMMON / SON / ANGLE(13), ANGLER(13), CPROB(13), DBLOS(13),
     A DBLOSL(13), DBLOSU(13), DBU(13), DELAY(13), DBW(13), NHP(13),
     B NYNOIS(13), PROBMP(13), RELIAB(13), SNDB(13), SNPR(13),
     C SNRLW(13), SNRUP(13), SPROB(13), VHIGH(13), MDL(16), RNEFF(13)
       CHARACTER MDL*1
      common /sncom/ SN90(13)
      common /cgains/ gaint(13),gainr(13)  ! transmitter & receiver gains
      COMMON /TON /ADJ, ADS, GNOS, GOT, PWRDB, REL, SL, SLS
     1, SPR, SU, SUS, XNOISE, ZNOISE, NF
      COMMON / allMODE /ABPS(20),CREL(20),FLDST(20),HN(20),HP(20),
     1PROB(20),RELY(20),RGAIN(20),SIGPOW(20),SN(20),
     2SPRO(20),TGAIN(20),TIMED(20),TLOSS(20),B(20),FSLOS(20),
     3grlos(20),adv(20),obf(20),
     CNMODE(20),TLLOW(20),TLHGH(20),EFF(20),NREL,NMMOD
      COMMON/INFORM/INFO,IHSHR,IHLNG
      DIMENSION TME(10)
      CHARACTER MODE*2, LAYTYP*2
      CHARACTER IMON*3, IRCVR*10, ITRAN*10, MODER*2, IRLAT*1,
     A IRLONG*1, ITLAT*1, ITLONG*1, NYEAR*5, LABEL*5, IEAST*1,
     B INORTH*1, ISOUTH*1, IWEST*1, LABLI*5, LABLJ*5, LABLK*5
C.....NORMAL DISTRIBUTION
      DATA TME/ 0.0,0.1257,0.2533, 0.3853,0.5244,0.6745,0.8416,1.0364,
     A 1.2815,1.6449/
      DATA XEPS/0.05/
C
```

```
C
C-----------------------------------------------------------------
C            Start of Subroutine
C-----------------------------------------------------------------
C
C
      INUM=NMMOD
C.....IF NO MODES RETURN.....................................
      IF(INUM.LE.0)RETURN
C.....BEGINNING OF RELIABILITY CALCULATION FOR EACH MODE
C.....USED TO SELECT 1 OF UP TO 20 MODES FOR EACH FREQ
      DU2=DU*DU
      DL2=DL*DL
      DO 310 IM = 1,INUM
      IF( HP(IM) - 70. ) 105,105,110
  105 CREL(IM) = 0.001
      RELY(IM) = 0.001
      GO TO 310
  110 IS = NMODE(IM)
      DSLF = TLLOW(IM)
      DSUF = TLHGH(IM)
C.....REQUIRED SIGNAL TO NOISE DISTRIBUTION
      D10R = SQRT(DL2 + DSLF*DSLF)
      D50R =  SN(IM)
      D90R = SQRT( DU2 + DSUF*DSUF )
      Z =  RSN - D50R
      IF( Z ) 290,290,295
  290 Z = Z/( D10R/1.28)
      GO TO 300
  295 Z = Z/(D90R/1.28)
  300 RELY(IM) = 1. - FNORML(Z)
C.....CREL IS NOT USED NOW
      CREL(IM) =  1000.
  310 CONTINUE
C
C.....END OF RELIABILITY CALCULATION FOR EACH MODE
C  MOST RELIABLE  MODE
C
      IR =1
      XREL = RELY(1)
      XHN  = HN(1)
      XSN  = SN(1)
C.....IF ONLY ONE MODE USE IT.................................
      IF(INUM.EQ.1)GO TO 145
      DO 140 IM= 2,INUM
C.....MAKE SELECTION BASED ON RELIABILITY FIRST BUT IF CLOSE SELECT ON
C.....LOWER NUMBER OF HOPS (IF THE NUMBER OF HOPS ARE EQUAL SELECT BY
C.....MEDIAN SNR)
      IF( ABS(RELY(IM) - XREL).LE.XEPS )THEN
C.......CLOSE SO TEST IF NUMBER OF HOPS ARE EQUAL.................
        IF(ABS(XHN - HN(IM)).LE.XEPS)THEN
C.........NUMBER OF HOPS ARE EQUAL SO TEST MEDIAN SNR.............
          IF( XSN.LT.SN(IM) )GO TO 139
        ELSE IF(XHN.GT.HN(IM))THEN
```

```
C.........THIS ONE HAS FEWER HOPS.....................................
          GO TO 139
        ENDIF
      ELSE IF(RELY(IM).GT.XREL)THEN
C.......THIS ONE IS MORE RELIABLE TRY IT..............................
        GO TO 139
      ENDIF
      GO TO 140
C.....THIS MODE IS BETTER SO TRY IT...................................
  139 IR = IM
      XHN=HN(IM)
      XSN=SN(IM)
      XREL = RELY(IM)
  140 CONTINUE
  145 CONTINUE
      NREL = IR
      IS = NMODE(IR)
      IF(INUM.eq.1)THEN
C.......ONLY ONE MODE SO SET MOST RELIABLE VALUES.....................
        RELIAB(IF) = RELY(IR)
        DBLOSL(IF) = TLLOW(IR)
        DBLOSU(IF) = TLHGH(IR)
        DBU(IF) = FLDST(IR)
        SNDB(IF) = SN(IR)
        DBW(IF) = SIGPOW(IR)
      ELSE
C.......ADD THE SIGNALS RANDOM PHASE i.e. ADD THE POWERS IN WATTS....
        XDSLW = 0.0
        XSIGS = 0.0
        XDSUP = 0.0
        XFLD = 0.
        DXSIGS=-1000.
        DXFLD=-1000.
        DXDSLW=-1000.
        DXDSUP=-1000.
        DO 369 IV=1,INUM
        DXSIGS=AMAX1(DXSIGS,SIGPOW(IV))
        DXFLD=AMAX1(DXFLD,FLDST(IV))
        DXDSLW=AMAX1(DXDSLW,SIGPOW(IV)-TLLOW(IV))
        DXDSUP=AMAX1(DXDSUP,SIGPOW(IV)+TLHGH(IV))
  369   CONTINUE
        DO 370 IM = 1,INUM
        ZEXP = .1*(SIGPOW(IM) - TLLOW(IM)-DXDSLW)
        XDSLW = XDSLW + 10.**ZEXP
        ZEXP = .1*(SIGPOW(IM)-DXSIGS)
        XSIGS = XSIGS +10.**ZEXP
        ZEXP = .1 * (SIGPOW(IM) + TLHGH(IM)-DXDSUP)
        XDSUP = XDSUP + 10.**ZEXP
C  MUST DO FIELD STRENGTH SEPARATE  BECAUSE OF RECEIVE ANTENNA
        ZEXP = .1*(FLDST(IM)-DXFLD)
        XFLD = XFLD + 10.**ZEXP
  370   CONTINUE
        SIGMED = DXSIGS+ 10.*ALOG10(XSIGS)
        DBLOSL(IF) = ABS( SIGMED - 10.*ALOG10(XDSLW)-DXDSLW )
```

```
        DBLOSU(IF) = ABS( DXDSUP + 10.*ALOG10(XDSUP)-SIGMED )
        DBW(IF) = SIGMED
        DELSIG = SIGMED - SIGPOW(IR)
        DBU(IF) = DXFLD + 10.*ALOG10(XFLD)
        SNDB(IF) = SN(IR) + DELSIG
C.......REDO RELIABILITY FOR SUM OF MODES.............................
        D10R = SQRT( DL2 + DBLOSL(IF)*DBLOSL(IF) )
        D50R = SNDB(IF)
        D90R = SQRT( DU2 + DBLOSU(IF)*DBLOSU(IF) )
        Z = RSN - D50R
        IF(Z.le.0.) then
           Z = Z/(D10R/1.28)
        else
           Z = Z/(D90R/1.28)
        end if
        RELIAB(IF) = 1. - FNORML(Z)
      ENDIF
      sn90(if)=d50r - d10r              !  lower decile (90%)
      gaint(if)=TGAIN(IR)               !  save transmitter gain
      gainr(if)=RGAIN(IR)               !  save receiver    gain
      SNRLW(IF) = D10R
      SNRUP(IF) = D90R
      ANGLE(IF) =  B(IR)
      VHIGH(IF)=HP(IR)
      DELAY(IF)=TIMED(IR)
      DBLOS (IF)= TLOSS(IR)
      CPROB(IF)=PROB(IR)
      MODE  (IF)= LAYTYP(IS)
      NHP   (IF)= HN(IR)
      NYNOIS(IF)= RCNSE
C------------    ADDED 9/24/91  (LONG PATH RCVR EFF CORRECTION)
      RNEFF(IF)=EFF(IR)
C  REQUIRED POWER GAIN  FOR SPECIFIED RELIABILITY.
      ITM = IABS( (LUFP - 50))/5 +1
      ITM = MINO(ITM,10)
      TMX = TME(ITM)/TME(9)
      IF( LUFP  - 50) 345,350,350
  345 SNPR(IF) = - (D50R + TMX * D90R) + RSN
      GO TO 355
  350 SNPR(IF) = - (D50R - TMX * D10R) + RSN
  355 CONTINUE
      IF(IAND(INFO,1).GT.0)THEN
        WRITE(99,'(29X,A,I4,A2)')'MOST RELIABLE MODE',NHP(IF),MODE(IF)
      ENDIF
      RETURN
      END
C-------------------------------
```

```
      SUBROUTINE SERPRB(SPROB)
C
C THIS SUBROUTINE CALCULATES THE SERVICE PROBABILITY WHICH IS THE
C PROBABILITY THAT THE GRADE OF SERVICE (GR) DEFINED BY THE REQUIRED
C SIGNAL-TO-NOISE RATIO WILL BE ACHIEVED OR EXCEEDED FOR A SPECIFIED
C TIME AVAILABILITY QR.  IN OTHER WORDS, THE PROBLEM IS TO ASSESS THE
C PROBABILITY THAT THE SPECIFIED GRADE OF SERVICE FOR A GIVEN
C PERCENTAGE OF THE DAYS WITHIN THE MONTH AT SOME SPECIFIED HOUR WILL
C BE PROVIDED BY THE CALCULATED AVAILABLE SIGNAL-TO-NOISE RATIO.
C
C INPUT:
C
C    LUFP       =    THE REQUIRED CIRCUIT RELIABILITY, WHICH IS AN
C                    ESTIMATE OF THE PERCENTAGE OF DAYS WITHIN THE
C                    MONTH THAT THE SIGNAL QUALITY WILL BE
C                    ACCEPTABLE, AND IS SPECIFIED AS TIME
C                    AVAILABILITY FOR SERVICE PROBABILITY
C                    CALCULATION. (NOTE:  LUFP IS EXPRESSED AS A
C                    PERCENTAGE.)  THE PROGRAM USES A DEFAULT VALUE
C                    OF 90 IF THIS PARAMETER IS NOT SET BY THE
C                    USER.
C    RSN        =    REQUIRED SIGNAL TO NOISE RATIO (DECRED)
C    SN(K)      =    Signal-to-Noise Ratio for MODE K (REGMOD, ESMOD)
C    HP(K)      =    Virtual Height of reflection for MODE K (RDMOD, ESMOD)
C    DU         =    UPPER DECILE OF THE TOTAL NOISE (GENOIS)
C    DL         =    LOWER DECILE OF THE TOTAL NOISE (GENOIS)
C    SIGM       =    Standard Deviation of the MEDIAN NOISE
C                    (prediction error) (GENOIS)
C    SIGU       =    Standard Deviation of the Noise UPPER DECILE
C                    (prediction error) (GENOIS)
C    SIGL       =    Standard Deviation of the Noise LOWER DECILE
C                    (prediction error) (GENOIS)
C    TLLOW(K)   =    Lower decile of TRANSMISSION LOSS DISTRIBUTION
C                    for mode K (REGMOD,ESMOD)
C    TLHGH(K)   =    Upper decile of TRANSMISSION LOSS DISTRIBUTION
C                    for mode K (REGMOD,ESMOD)
C    ADS        =    Standard deviation of MEDIAN EXCESS SYSTEM LOSS
C                    (prediction error) (SIGDIS)
C    SUS        =    Standard deviation of UPPER DECILE of EXCESS SYSTEM LOSS
C                    (prediction error) (SIGDIS)
C    SLS        =    Standard deviation of LOWER DECILE of EXCESS SYSTEM LOSS
C                    (prediction error) (SIGDIS)
C
C
C OUTPUT:
C    SPRO(K)    =    Service Probability for mode K for the required
C                    reliability
C    SPROB      =    Maximum of SPRO(K) for all modes K=1,NMMOD.
C    D50S       =    Median Signal-to-Noise ratio deviation of best mode.
C    D10S       =    "Additional" required Signal-to-Noise ratio
C                    for the required reliability
C
C LOCAL VARIABLES
C         DR    =    Standard Deviation of the Required Signal-to-Noise
```

```
C                   Ratio (RSN)
C              =    2.0 dB (SET IN DATA STATEMENT)
C        TMX   =    Standard normal deviate corresponding to the
C                   required circuit reliability (LUFP)
C                   NORMAL DISTRIBUTION
C                   .5 = 0.0,  .55 = 0.1257,..., .9 = 1.2815, .95
C                   = 1.6449
C                   The selection is made on the nearest 5% value of LUFP.
C   FNORML(Z) =     Standard normal deviate cumulative distribution function
C-------------------------------------------------------------
C          Start of Common Area
C-------------------------------------------------------------
C
      COMMON/ANOIS/ATNU,ATNY,CC,TM,RCNSE,DU,DL,SIGM,SIGU,SIGL,KJ,JK
      COMMON/CON/D2R,DCL,GAMA,PI,PI02,R2D,RZ,VOFL
      COMMON /DON /ALATD, AMIN, AMIND, BTR, BTRD, DLONG, DMP, ERTR, GCD,
     1 GCDKM, PMP, PWR, RLAT, RLATD, RLONG, RLONGD, RSN, SIGTR, TLAT,
     2 TLATD,TLONG,TLONGD,BRTD,FLUX,SSN,D90R,D50R,D10R,D90S,D50S,D10S
      COMMON / ION / IHRE, IHRO, IHRS, LUFP, METHOD, NOISE, NPSL
      COMMON/TON/ADJ,ADS,GNOS,GOT,PWRDB,REL,SL,SLS,SPR,SU,SUS
     A ,XNOISE,ZNOISE,NF
      COMMON / allMODE /ABPS(20),CREL(20),FLDST(20),HN(20),HP(20),
     1PROB(20),RELY(20),RGAIN(20),SIGPOW(20),SN(20),
     2SPRO(20),TGAIN(20),TIMED(20),TLOSS(20),B(20),FSLOS(20),
     3grlos(20),adv(20),obf(20),
     CNMODE(20),TLLOW(20),TLHGH(20),EFF(20),NREL,NMMOD
      DIMENSION TME(10),d10sa(20),d50sa(20)
C.....NORMAL DISTRIBUTION
      DATA TME/0.0,0.1257,0.2533,0.3853,0.5244,0.6745,0.8416,1.0364
     A , 1.2815, 1.6449 /
C.....DR IS THE PREDICTION ERROR IN RSN, REQUIRED SNR
      DATA DR/2./
C-------------------------------------------------------------------
C              Start of Subroutine
C-------------------------------------------------------------------
C
      SPR = 0.0
C.....LUFP IS THE REQUIRED RELIABILITY
      ITM = IABS( LUFP - 50) /5 +1
      ITM = MINO(ITM,10)
      TMX = TME(ITM)
C.....BEGINNING OF FINDING THE SERVICE PROBABILITY FOR EACH MODE
      DO 145 K = 1,NMMOD
      IF( HP(K).le.70.)  then
C.....SET DEFAULT FOR NO MODE
         SPRO(K) = .001
         d50sa(K)=0.
         d10sa(K)=0.
         GO TO 145
      end if
      IF( LUFP.ge.50) then
C.....DN IS THE NOISE
C.....DS IS THE SIGNAL FOR THIS MODE
         DN  = TMX *DU/DCL
```

```
            DS  = TMX *TLLOW(K)/DCL
            XLH = -1.
C.....SIGNAL PREDICTION ERROR
            DSO = TMX *SUS
            DNO = TMX * SIGU/DCL
         else
            DN  = TMX *DL/DCL
            DS  = TMX *TLHGH(K)/DCL
            XLH = 1.
            DSO = TMX * SLS
            DNO = TMX *SIGL/DCL
         end if
C.....D1OS IS THE LOWER DECILE
C.....D5OS IS THE MEDIAN
C.....D9OS IS THE UPPER DECILE
         D5OSa(K)= SQRT(DN*DN + DS*DS)
         D1OSa(K)= D5OSa(K) +
        +          SQRT(SIGM*SIGM + ADS*ADS + DNO*DNO + DSO*DSO + DR*DR)
         D5OSa(K) =  SN(K)  + XLH*D5OSa(K)
         Z = (RSN - D5OSa(K))/D1OSa(K)
C.....SERVICE PROBABILITY FOR THIS MODE
         SPRO(K) =  1. - FNORML(Z)
  145 CONTINUE
C.....END OF FINDING THE SERVICE PROBABILITY FOR EACH MODE
C
C.....USE THE MAXIMUM
         imax=1
         do 150 K=2,NMMOD
         if(SPRO(K).gt.SPRO(imax)) imax=K
150      continue
         SPROB = SPRO(imax)
         D1OS=d1Osa(imax)
         D5OS=d5Osa(imax)
         D9OS=d1Osa(imax)
         RETURN
         END
C-------------------------------
```

*U.S. GOVERNMENT PRINTING OFFICE: 1993-774-025/89037

FORM NTIA-29
(4-80)

U.S. DEPARTMENT OF COMMERCE
NAT'L. TELECOMMUNICATIONS AND INFORMATION ADMINISTRATION

BIBLIOGRAPHIC DATA SHEET

	1. PUBLICATION NO.	2. Gov't Accession No.	3. Recipient's Accession No.
	NTIA Report 93- 297		

4. TITLE AND SUBTITLE
A Critique of the Reliability & Service Probability Calculations for the Ionospheric Communication Analysis and Prediction Program - IONCAP

5. Publication Date
August 1993

6. Performing Organization Code
NTIA ITS.S3

7. AUTHOR(S)
A. D. Spaulding and F.G. Stewart

8. PERFORMING ORGANIZATION NAME AND ADDRESS
National Telecommunications & Information Administration
Institute for Telecommunication Sciences
325 Broadway
Boulder, CO 80303

9. Project/Task/Work Unit No.

10. Contract/Grant No.

11. Sponsoring Organization Name and Address
U.S. Information Agency
Bureau of Broadcasting
330 Independence Avenue
Washington, DC 20547

12. Type of Report and Period Covered

13.

14. SUPPLEMENTARY NOTES

15. ABSTRACT *(A 200-word or less factual summary of most significant information. If document includes a significant bibliography or literature survey, mention it here.)*

This report presents an analysis and explanation of the system performance calculations performed by the reliability and service probability subroutines of the Ionospheric Communications Analysis and Prediction Program "IONCAP" and an IONCAP derivative termed "VOACAP". A review of the three components needed to properly statistically describe the performance of a communications system or link is presented. These components are needed to account for the short-term and long-term statistical variations of the desired signal and the interference and also the inherent prediction errors. A detailed numerical example is given for illustration and explanation. The functioning of the subroutines that perform the system performance calculations is documented in detail followed by suggestions for modification and improvement.

16. Key Words *(Alphabetical order, separated by semicolons)*

IONCAP, VOACAP, reliability, noise, overall operating noise threshold, system performance

17. AVAILABILITY STATEMENT

☒ UNLIMITED.

☐ FOR OFFICIAL DISTRIBUTION.

18. Security Class. *(This report)*
Unclassified

19. Security Class. *(This page)*

20. Number of pages
47

21. Price